


## NTSC Video Using the 68HC908QY4

### Abstract

The Motorola 68HC908QY4 is used to produce an NTSC video signal which can be used in several applications. By using 28.63636 MHz for the clock, the QY4 bus speed is 7.15909 MHz. which is twice the chroma frequency of 3.579545 MHz. The Interrupt Timer is used to generate an interrupt every 455 bus cycles resulting in an interrupt rate of 15.734 KHz. which precisely the frequency of NTSC sync. Because interrupts that occur in the middle of a multi-cycle instruction are not recognized until the instruction is completed, the main line consists of a table of NOPs; all of the sync functions are performed during the interrupt, and the interrupt routine is returned to the start of the NOP table.

One application using the QY4 to produce NTSC sync is an NTSC Test Pattern Generator which produces simple test patterns in order to evaluate a TV's geometric distortion, high voltage regulation, and interlace quality. Another is a VCR Pacifier to allow a VCR to record an audio-only signal such as from a radio. The VCR Pacifier also includes a real-time clock which is displayed either in video or by using the TV's closed captioned decoder. Low-bandwidth data from the 68HC908QY4's A/D inputs can be displayed the same way, turning the VCR into a two-channel data logger. As a bonus, a Video Line Trigger is presented which counts video lines from an external NTSC video source and produces a pulse to trigger an oscilloscope at the selected line(s).



VCR Pacifier – NTSC Video with Audio Inputs

Sample Code

The following code produces Horizontal Sync for Field 1 active video.

```

=====
* Active Video: Field 1 begins on Line 10, ends on Line 262 1/2
* Real Horizontal Sync
TIRQ3:

* HBLANK signal will be about 11 us total
  BCLR  HBLANK,PORTB          ; [4]  HBLANK on

* HSYNC front porch (1.27 us = 9 cycles total)
  BCLR  HBLANK,PORTB          ; [4]  waste cycles
  NOP                    ; [1]

*-----
* HSYNC will be 4.77 us = 34 cycles
  BSET  CSYNC,PORTB          ; [4]  SYNC on

  BCLR  TOF,TSC              ; [4]  reset interrupt flag

  LDX   #8                    ; [2]
tq3a:  DBNZX tq3a              ; [3]
  BCLR  CSYNC,PORTB          ; [4]  SYNC off
* -----
* Color Burst Front Porch Gate 0.38 us (instruction fetch time is sufficient)

* Color Burst: at least 8 cycles of 3.579545 MHz = 2.3 us = 16 cycles
* We will do 9 cycles of 3.579545 MHz = 18 cycles
  BSET  CBGATE,PORTB         ; [4]
  BSET  CBGATE,PORTB         ; [4] waste cycles
  BSET  CBGATE,PORTB         ; [4] waste cycles
  BSET  CBGATE,PORTB         ; [4] waste cycles
  NOP                    ; [1]
  NOP                    ; [1]
  BCLR  CBGATE,PORTB         ; [4]
*-----
* From start of HSYNC to end of HBLANK is 9.22 us = 66 cycles
* We need 6 more cycles
  BCLR  CBGATE,PORTB         ; [4] waste cycles
  NOP                    ; [1]
  NOP                    ; [1]
  BSET  HBLANK,PORTB         ; [4]  HBLANK off
*-----
* Check Vertical Count
  DBNZ  vcount,tq3c

* We have reached the end of active video. Next, start VBI
  LDHX  #TIRQ4                ; Timer Overflow Vector: Start VBI next time
  STHX  timer_vec

  BSET  VBLANK,PORTB         ; oscilloscope trigger high
  JMP   TIRQRET

* Video is still active
* We have about 50 us to do something else
tq3c:  JSR   Do_Field1         ; draw a box
  JMP   TIRQRET
=====

```

Block Diagram